Supplementary Material

Evaluating Diagnostic Tests and Quantifying Prevalence for Tropical Infectious Diseases: a Paradigm of Latent Class Modelling Approaches with and without a Gold Standard for Schistosomiasis Diagnosis

1 Fit Measures: Univariate and bivariate GF-Fits for the LC, LCRE and FM models for the Uganda Data set

Table 1 gives the GF-Fit values for the pair POC-CCA and SEA-ELISA diagnostic tests and for the LC, LCRE and FM models for the Uganda data set.

Table 1: Contribution to GF-Fits: POC-CCA by SEA-ELISA, N=258

POC-CCA	SEA-ELISA	$N \times p_{ab}$	$N \times \hat{p}_{ab}$ LC	$N \times \hat{p}_{ab}$ LCRE	$N \times \hat{p}_{ab}$ FM
0	0	59	50.14	50.21	55.03
0	1	52	61.30	61.10	56.69
1	0	5	14.91	14.20	10.68
1	1	142	131.66	132.49	135.60
Total GF-Fit			10.37	9.54	4.00

Table 2 gives the univariate and bivariate GF-Fits for the LC, LCRE and FM models for all pairs of diagnostic tests. The univariate fits are in the diagonal and the bivariate fits are below the diagonal. Among the three models the LCRE provides the smallest bivariate GF-Fits but all models show adequate fit. Each value that appears in the table is smaller than 16.

Table 2: Univariate and Bivariate GF-Fits, Uganda dataset

LC

	POC-CCA	DNA-TaqMan	SEA-ELISA	Kato-Katz1	Kato-Katz2
POC-CCA	0.00				
DNA-TaqMan	0.60	0.00			
SEA-ELISA	10.37	3.48	0.02		
Kato-Katz1	1.66	0.51	0.22	0.00	
Kato-Katz2	0.58	0.13	0.13	2.49	0.00

LCRE

	POC-CCA	DNA-TaqMan	SEA-ELISA	Kato-Katz1	Kato-Katz2
POC-CCA	0.00				
DNA-TaqMan	0.32	0.00			
SEA-ELISA	9.54	3.09	0.00		
Kato-Katz1	0.58	0.16	0.01	0.00	
Kato-Katz2	0.02	0.00	0.12	1.59	0.00

FM

	POC-CCA	DNA-TaqMan	SEA-ELISA	Kato-Katz1	Kato-Katz2
POC-CCA	0.01				
DNA-TaqMan	0.35	0.01			
SEA-ELISA	4.00	3.11	0.06		
Kato-Katz1	1.99	3.14	0.44	0.03	
Kato-Katz2	0.92	2.16	0.07	11.78	0.03

2 Figures: Figures 1 and 2 for Bias and MSE respectively of estimated sensitivities and specificities for the simulation scenario of sample size=250 and prevalence=0.4.

Figure 1: Bias of parameter estimates (sensitivities and specificities) as estimated by the LC, LCRE and FM models under differing proportions of missing gold standard, and under differing data generating mechanisms (sample size = 250, prevalence = 0.4).



Figure 2: MSE of parameter estimates (sensitivities and specificities) as estimated by the LC, LCRE and FM models under differing proportions of missing gold standard, and under differing data generating mechanisms (sample size = 250, prevalence = 0.4).



3 JAGS code: Code for fitting the latent class model under conditional independence, the latent class model with Gaussian random effects and the finite mixture latent class model.

Latent Class Model with Conditional Independence

```
model{
for (i in 1:n) {
  d[i] ~ dbern(tau)
  for (j in 1:p) {
    pi.x[i, j] <- d[i] * pi[j,1] + (1 - d[i]) * pi[j,2]
    X[i, j] ~ dbern(pi.x[i, j])
  }
}

# Priors (uninformative)
tau ~ dbeta(1,1)
for (j in 1:(p-1)) {
  pi[j,1] ~ dbeta(1,1)
  pi[j,2] ~ dbeta(1,1)
}
pi[p, 1] <- 1  # This fixes sens and spec to 1.If no gold standard
pi[p, 2] <- 1  # then assign dbeta(1,1) priors.

# Sensitivities and specificities
for (j  in 1:p) {
  sens[j] <- pi[j,1]
  spec[j] <- 1 - pi[j,2]
}
}

#data# X, n, p
#monitor# tau, sens, spec, deviance
```

Latent class Gaussian random effects model

```
model{
  for (i in 1:n) {
    d[i] ~ dbern(tau)
    for (j in 1:p) {
      u1[i,j] ~ dnorm(beta[j,1], psi[1])
      u0[i,j] ~ dnorm(beta[j,2], psi[2])
      pi.x[i,j] <- d[i] * phi(u1[i,j]) + (1 - d[i]) * phi(u0[i,j])
      X[i,j] ~ dbern(pi.x[i,j])
    }
  }

  # Priors
  tau ~ dbeta(1,1)
  for (j in 1:(p-1)) {
    beta[j,1] ~ dnorm(0,0.01)
    beta[j,2] ~ dnorm(0,0.01)
  }
  beta[p,1] <- 100   # This fixes sens and spec to 1. Assign
  beta[p,2] <- -100  # dnorm(0,0.01) prior if no gold std.
  for (k in 1:2) {
    psi[k] ~ dgamma(0.1, 0.1)
    sigma[k] <- 1 / sqrt(psi[k])
  }

  # Sensitivities and specificities
  for (j  in 1:p) {
    sens[j] <- phi(beta[j,1] / sqrt(1 + pow(sigma[1], 2)))
    spec[j] <- 1 - phi(beta[j,2] / sqrt(1 + pow(sigma[2], 2)))
  }
}

#data# X, n, p
#monitor# tau, sens, spec, beta, deviance
```

Finite Mixture Latent Class Model

```
model{
  # NOTE: class 1 = diseased, class 2 = healthy
  for (i in 1:I) {  # individuals
    for (j in 1:J) {  # items
      Y[i,j] ~ dbern(A[i,j])
      A[i,j] <- pow(w[j, 2] * d[i] + (1 - w[j,1]) * (1 - d[i]), //
      1 - l[i]) - l[i] * (1 - d[i])
    }
    l[i] ~ dbern(pi.l[i])
    pi.l[i] <- (1 - d[i]) * eta[1] + d[i] * eta[2]
    d[i] ~ dbern(tau)
  }

  # Priors and sens/spec
  tau ~ dbeta(1,1)
  for (k in 1:K) {
    eta[k] ~ dbeta(1,1)
    w[J,k] <- 1  # if no gold std, assign dbeta(1,1) prior
  }
  for (j in 1:(J-1)) {
    w[j,1] ~ dbeta(1,1)
    w[j,2] ~ dbeta(1,1)
    sens[j] <- eta[2] + (1 - eta[2]) * w[j,2]
    spec[j] <- eta[1] + (1 - eta[1]) * w[j,1]
  }
  sens[J] <- eta[2] + (1 - eta[2]) * w[J,2]
  spec[J] <- eta[1] + (1 - eta[1]) * w[J,1]
}

#data# I, J, K, Y
#monitor# tau, sens, spec, eta, deviance
```